

INTEGRAL INEQUALITIES AND COMPUTER ALGEBRA SYSTEMS

J. ROUMELIOTIS¹

SCHOOL OF COMPUTER SCIENCE AND MATHEMATICS
VICTORIA UNIVERSITY OF TECHNOLOGY
PO BOX 14428, MELBOURNE VIC 8001, AUSTRALIA.

John.Roumeliotis@vu.edu.au
URL: <http://www.staff.vu.edu.au/johnr>

Received 27 May, 2005; accepted 01 December, 2005

Communicated by I. Pinelis

ABSTRACT. Theoretical results involving approximation of integrals are often established from the construction and resultant manipulation of an appropriate kernel. The systematic use of these kernels has produced an abundance of new approximations and error estimates in terms of norms of the integrand. Notwithstanding the great success of this approach, many approximations and error results have yet to be discovered due to the algebraic complexities involved; especially those that involve product integrands.

We outline a method that uses the computer algebra system Maple that is able to recapture the well known Ostrowski, trapezoidal and Simpson's inequalities. Moreover, the technique, which involves manipulation of the Peano kernel, can be adapted to develop new rules.

Key words and phrases: Symbolic computation, Numerical integration, Simpson's rule, Trapezoidal rule, Integral inequalities.

2000 *Mathematics Subject Classification.* Primary: 68W30, 65D30; Secondary: 26D10, 26D15.

1. INTRODUCTION

Newton-Cotes and Newton-Cotes type integration has been an area of much recent activity. Cerone, Dragomir, Pečarić and others have reported on generalizations of the Ostrowski, trapezoidal and Simpson rules. These results have involved new bounds being expressed in different norms [1, 7, 9, 14, 16] as well as new generalizations of the rules themselves [12, 15]. For a good overview we refer the reader to the survey [1] and the book [11]. Much may be garnered from the work of Cerone [2, 3, 4, 5, 6] whose results are a considerable generalization of previous work.

ISSN (electronic): 1443-5756

© 2005 Victoria University. All rights reserved.

This paper is based on the talk given by the author within the "International Conference of Mathematical Inequalities and their Applications, I", December 06-08, 2004, Victoria University, Melbourne, Australia [<http://rgmia.vu.edu.au/conference>]

¹This author is currently employed with the National Australia Bank. Contact email is: John_Roumeliotis@national.com.au.

Most of these results can be obtained via manipulation of the appropriate Peano kernel and nearly all of the algebra required to establish these results require partial integration, differentiation and some simplification - all of which may be done by modern computer algebra systems (CAS's) such as Maple [13], Mathematica and MuPad. The algebraic techniques employed here are inspired by the early work of Dragomir and Wang [8].

In this paper we begin by detailing the manual steps required to prove the Ostrowski inequality. In Sections 3 and 4 we show, with considerable detail, how Maple can be used to prove the Ostrowski and corrected trapezoidal inequalities. In Section 5, Maple is used to establish a new corrected trapezoidal rule whose upper bound is 1.7 times smaller than the canonical corrected trapezoidal rule. In Section 6, Maple is used to recapture a recently published modified Simpson rule [17]. Each of Sections 3 – 6 are independent Maple worksheets which the reader is encouraged to download from <http://www.roumeliotis.com.au/john/maple>.

2. OSTROWSKI'S INEQUALITY – MANUAL CALCULATION

Ostrowski's inequality, a generalized mid-point inequality, is quite simple to establish. The proof involves partial integration of a piecewise linear Peano kernel to obtain the rule and integration of the absolute value of the kernel to obtain the bound.

The proof of Ostrowski's inequality is well known and has been reproduced many times in the literature. See for example the book [11] or the paper [10].

Theorem 2.1 (Ostrowski). *Let $f : I \subseteq \mathbb{R} \rightarrow \mathbb{R}$ be a differentiable mapping in I° (I° is the interior of I), and let $a, b \in I^\circ$ with $a < b$. If $f' : (a, b) \rightarrow \mathbb{R}$ is bounded on (a, b) , i.e., $\|f'\|_\infty := \sup_{t \in (a, b)} |f'(t)| < \infty$, then we have the inequality:*

$$(2.1) \quad \left| \int_a^b f(t) dt - (b-a)f(x) \right| \leq \left[\frac{(b-a)^2}{4} + \left(x - \frac{a+b}{2} \right)^2 \right] \|f'\|_\infty$$

for all $x \in (a, b)$.

Proof. Let the Peano kernel p be given by

$$p(t) = \begin{cases} t - a, & a \leq t \leq x, \\ t - b, & x < t \leq b, \end{cases}$$

for some $x \in [a, b]$. Integrating the product $p(t)f'(t)$ gives

$$(2.2) \quad I = \int_a^b p(t)f'(t) dt$$

$$(2.3) \quad = \int_a^x p(t)f'(t) dt + \int_x^b p(t)f'(t) dt$$

$$(2.4) \quad = (x-a)f(x) - \int_a^x f(t) dt - (x-b)f(x) - \int_x^b f(t) dt$$

$$(2.5) \quad = (b-a)f(x) - \int_a^b f(t) dt.$$

Hölder's inequality is used to obtain the bound.

$$\begin{aligned}
 (2.6) \quad & \left| \int_a^b f(t) dt - (b-a)f(x) \right| = |I| \\
 (2.7) \quad & = \left| \int_a^b p f' dt \right| \\
 (2.8) \quad & \leq \|f'\|_\infty \int_a^b |p| dt \\
 (2.9) \quad & = \|f'\|_\infty \left(\int_a^x t-a dt + \int_x^b b-t dt \right) \\
 (2.10) \quad & = \left[\frac{(b-a)^2}{4} + \left(x - \frac{a+b}{2} \right)^2 \right] \|f'\|_\infty.
 \end{aligned}$$

From a computer algebra perspective, the important steps are the integration by parts in (2.3) to (2.4) and integration of the positive kernel in (2.8) to (2.10). For complex kernels, it is often difficult to ascertain whether a kernel is positive (as in equation (2.9)), and certainly cannot be automatically determined by a CAS. □

In the next section we show the Maple steps used to establish (2.1). It is helpful to compare each step with those in (2.2)–(2.10) above.

3. OSTROWSKI INEQUALITY

In this section, we will use the Maple computer algebra system to derive the well known Ostrowski inequality. This section is an actual Maple worksheet called **ostrowski.mws** which you can download from <http://www.roumeliotis.com.au/john/maple>.

```

> restart;
> with( student ):

```

The Ostrowski inequality is a first order approximation for a finite integral using a general interior point x . Hence the kernel is of first order and vanishes at the points $t = a$ and $t = b$. Define the general kernel $p(t, x) = t - x$ where $a \leq x$ and $x \leq b$

```

> p := (t,x) -> t-x;
      p := (t, x) -> t - x

```

Integrating the product $p(t) (\frac{d}{dt} f(t))$ by parts and simplifying produces the approximation.

```

> Ostr_rule := intparts( Int(
> p(t,a)*D(f)(t) , t=a..x ) , p(t,a) ) +
> intparts( Int( p(t,b) * D(f)(t) , t=x..b ) , p(t,b) );

```

$$Ostr_rule := (x-a)f(x) - \int_a^x f(t) dt - (x-b)f(x) - \int_x^b f(t) dt$$

```

> Ostr_rule := collect( combine( Ostr_rule ) , f );

```

$$Ostr_rule := (b-a)f(x) + \int_a^b -f(t) dt$$

To obtain the bound, via Hölder's inequality, we need to integrate $|p(t)|$.

```

> Ostr_bound := int( p(t,a) , t=a..x ) - int( p(t,b) ,
> t=x..b );

```

$$Ostr_bound := x^2 - \frac{a^2}{2} - a(x-a) - \frac{b^2}{2} + b(-x+b)$$

Completing the square and factorizing the constant term produces the result

```
> Ostr_bound := op(1, completesquare(Ostr_bound , x ))
> + factor( Ostr_bound - op(1, completesquare(
> Ostr_bound , x )) );
```

$$Ostr_bound := \left(x - \frac{a}{2} - \frac{b}{2}\right)^2 + \frac{(a-b)^2}{2}$$

4. CORRECTED TRAPEZOIDAL RULE

In this section, we will use the Maple computer algebra system to derive the corrected trapezoidal rule using a Peano kernel approach. This section is an actual Maple worksheet called **corrected_trap.mws** which you can download from <http://www.roumeliotis.com.au/john/maple>.

The corrected rule is fourth order and is defined over one interval $[a, b]$, this contrasts with the Ostrowski kernel which has separate definitions for $[a, x]$ and $(x, b]$. Recall that the corrected trapezoidal rule is

$$(4.1) \quad \int_a^b f(x) dx = \frac{(b-a)f(a)}{2} + \frac{(b-a)f(b)}{2} + \frac{(b-a)^2 D(f)(a)}{12} - \frac{(b-a)^2 D(f)(b)}{12} + E$$

where

$$|E| \leq \frac{\max(|(D^{(4)}(f))|(b-a)^5)}{720}.$$

Define an arbitrary fourth order polynomial as our Peano kernel and solve for the coefficients of the kernel by specifying the required coefficients of the rule. To begin, restart the Maple kernel and load the student package which is required for the following calculations.

```
> restart; with( student );
```

Define the general fourth order Peano kernel, $p(t)$.

```
> p := t -> 1/4!*(t-A[1])*(t-A[2])*(t-A[3])*(t-A[4]);
p := t -> \frac{(t-A_1)(t-A_2)(t-A_3)(t-A_4)}{4!}
```

Define I_rule to be the integral of the product $(D^{(4)}(f)(t)p(t)$.

```
> I_rule := Int( (D@@4)(f)(t) * p(t) , t=a..b );
```

$$I_rule := \int_a^b \frac{1}{24} (D^{(4)}(f)(t) (t-A_1)(t-A_2)(t-A_3)(t-A_4) dt$$

This is integrated by parts four times. The Maple command `applyop` is used to integrate the required part of the expression.

```
> intparts( I_rule , p(t) );
> applyop( intparts , nops(%) , % , diff( p(t), t ) );
> applyop( intparts , nops(%) , % , diff( p(t), t$2 ) );
> Corr_Rule := applyop( intparts , nops(%) , % , diff( p(t), t$3 ) );
```

$$\begin{aligned}
 \text{Corr_Rule} := & \frac{1}{24} (b - A_1) (b - A_2) (b - A_3) (b - A_4) (D^{(3)})(f)(b) \\
 & - \frac{1}{24} (a - A_1) (a - A_2) (a - A_3) (a - A_4) (D^{(3)})(f)(a) - (\\
 & \frac{1}{24} (b - A_2) (b - A_3) (b - A_4) + \frac{1}{24} (b - A_1) (b - A_3) (b - A_4) \\
 & + \frac{1}{24} (b - A_1) (b - A_2) (b - A_4) + \frac{1}{24} (b - A_1) (b - A_2) (b - A_3))(D^{(2)})(f)(b) + (\\
 & \frac{1}{24} (a - A_2) (a - A_3) (a - A_4) + \frac{1}{24} (a - A_1) (a - A_3) (a - A_4) \\
 & + \frac{1}{24} (a - A_1) (a - A_2) (a - A_4) + \frac{1}{24} (a - A_1) (a - A_2) (a - A_3))(D^{(2)})(f)(a) + (\\
 & \frac{1}{12} (b - A_3) (b - A_4) + \frac{1}{12} (b - A_2) (b - A_4) + \frac{1}{12} (b - A_2) (b - A_3) \\
 & + \frac{1}{12} (b - A_1) (b - A_4) + \frac{1}{12} (b - A_1) (b - A_3) + \frac{1}{12} (b - A_1) (b - A_2))D(f)(b) - (\\
 & \frac{1}{12} (a - A_3) (a - A_4) + \frac{1}{12} (a - A_2) (a - A_4) + \frac{1}{12} (a - A_2) (a - A_3) \\
 & + \frac{1}{12} (a - A_1) (a - A_4) + \frac{1}{12} (a - A_1) (a - A_3) + \frac{1}{12} (a - A_1) (a - A_2))D(f)(a) \\
 & - (b - \frac{1}{4} A_4 - \frac{1}{4} A_3 - \frac{1}{4} A_2 - \frac{1}{4} A_1) f(b) + (a - \frac{1}{4} A_4 - \frac{1}{4} A_3 - \frac{1}{4} A_2 - \frac{1}{4} A_1) f(a) \\
 & + \int_a^b f(t) dt
 \end{aligned}$$

To re-capture the corrected trapezoidal integration rule, we solve for

A_1, A_2, A_3, A_4

such that the coefficients of $\frac{d^2}{dt^2} f(t)$ and $\frac{d^3}{dt^3} f(t)$ vanish at $t = a$ and $t = b$.

> eq1 := coeff(Corr_Rule , (D@@2)(f)(a)) = 0;

$$\begin{aligned}
 \text{eq1} := & \frac{1}{24} (a - A_2) (a - A_3) (a - A_4) + \frac{1}{24} (a - A_1) (a - A_3) (a - A_4) \\
 & + \frac{1}{24} (a - A_1) (a - A_2) (a - A_4) + \frac{1}{24} (a - A_1) (a - A_2) (a - A_3) = 0
 \end{aligned}$$

> eq2 := coeff(Corr_Rule , (D@@2)(f)(b)) = 0;

$$\begin{aligned}
 \text{eq2} := & -\frac{1}{24} (b - A_2) (b - A_3) (b - A_4) - \frac{1}{24} (b - A_1) (b - A_3) (b - A_4) \\
 & - \frac{1}{24} (b - A_1) (b - A_2) (b - A_4) - \frac{1}{24} (b - A_1) (b - A_2) (b - A_3) = 0
 \end{aligned}$$

> eq3 := coeff(Corr_Rule , (D@@3)(f)(a)) = 0;

$$\text{eq3} := -\frac{1}{24} (a - A_1) (a - A_2) (a - A_3) (a - A_4) = 0$$

> eq4 := coeff(Corr_Rule , (D@@3)(f)(b)) = 0;

$$\text{eq4} := \frac{1}{24} (b - A_1) (b - A_2) (b - A_3) (b - A_4) = 0$$

> solve({eq1,eq2,eq3,eq4} , {A[1],A[2],A[3],A[4]});

$$\{A_4 = a, A_3 = b, A_1 = b, A_2 = a\}, \{A_4 = a, A_3 = b, A_2 = b, A_1 = a\},$$

$$\{A_4 = a, A_1 = b, A_2 = b, A_3 = a\}, \{A_1 = b, A_2 = a, A_3 = a, A_4 = b\},$$

$$\{A_2 = b, A_1 = a, A_3 = a, A_4 = b\}, \{A_3 = b, A_2 = a, A_1 = a, A_4 = b\}$$

Without loss of generality we can choose the first solution and we assign the variables A_1, A_2, A_3, A_4 to this.

```
> assign( %[1] );
```

The integration rule is

```
> Corr_Rule;
```

$$\frac{1}{12} (b-a)^2 D(f)(b) - \frac{1}{12} (-b+a)^2 D(f)(a) - \left(\frac{b}{2} - \frac{a}{2}\right) f(b) + \left(\frac{a}{2} - \frac{b}{2}\right) f(a) + \int_a^b f(t) dt$$

To obtain the error bound, we employ a Hölder inequality

$$\int_a^b |p(t) (D^{(4)})(f)(t)| dt \leq \max(|(D^{(4)})(f)|) \int_a^b |p(t)| dt$$

which requires calculation of $|p|$. It is easy to see that $0 < p$

```
> p(t);
```

$$\frac{(t-b)^2 (t-a)^2}{24}$$

so calculation of the bound is a simple matter of integration

```
> int( p(t) , t=a..b );
```

$$\frac{b^5}{120} - \frac{a^5}{120} + \frac{\left(-\frac{b}{12} - \frac{a}{12}\right) (b^4 - a^4)}{4} + \frac{\left(\frac{1}{24} b^2 + \frac{1}{6} b a + \frac{1}{24} a^2\right) (b^3 - a^3)}{3}$$

$$+ \frac{\left(-\frac{1}{12} b^2 a - \frac{1}{12} a^2 b\right) (b^2 - a^2)}{2} + \frac{b^2 a^2 (b-a)}{24}$$

```
> factor( % );
```

$$-\frac{(-b+a)^5}{720}$$

5. NEW CORRECTED TRAPEZOIDAL RULE

In this section, we will use the Maple computer algebra system to derive a new corrected trapezoidal rule using a Peano kernel approach. This section is an actual Maple worksheet called **new_corrected_trap.mws** which you can download from <http://www.roumeliotis.com.au/john/maple>.

The corrected trapezoidal rule in Section 4 was obtained by determining the coefficients of the Peano kernel so that the $\frac{d^2}{dt^2} f(t)$ and $\frac{d^3}{dt^3} f(t)$ terms vanish in the integration rule. The $\frac{d}{dt} f(t)$ terms are equal in magnitude and opposite in sign, so that they cancel everywhere except at the boundary in a composite rule. There is an opportunity to create a further correction to the corrected trapezoidal rule by imposing the condition that all three derivatives cancel for a composite rule. This would leave one free parameter which can be used to minimize the error bound. Care must be taken when integrating the positive kernel to produce the error bound, since determining the sign of the kernel cannot be automated.

```
> restart;
```

```
> with(student):
```

Call in special plotting packages for 3D plots to be produced in this section.

```
> with( plots ): with( plottools ):
```

Warning, the name changecoords has been redefined

Warning, the assigned name arrow now has a global binding

As with the corrected rule, the kernel is a fourth order polynomial since this rule will have fourth order accuracy.

```
> p := t->1/4!*(t-A[1])*(t-A[2])*(t-A[3])*(t-A[4]);
```

$$p := t \rightarrow \frac{(t - A_1)(t - A_2)(t - A_3)(t - A_4)}{4!}$$

Define *Corr_Rule* to be the integral of the product $(D^{(4)})(f)(t)p(t)$.

```
> Corr_Rule := Int( (D@@4)(f)(t) * p(t) , t=a..b );
```

$$Corr_Rule := \int_a^b \frac{1}{24} (D^{(4)})(f)(t) (t - A_1)(t - A_2)(t - A_3)(t - A_4) dt$$

Integrate by parts four times.

```
> intparts( Corr_Rule , p(t) );
```

```
> applyop( intparts , nops(%) , % , diff( p(t),t ) );
```

```
> applyop( intparts , nops(%) , % , diff( p(t),t$2 ) );
```

```
> Corr_Rule := applyop( intparts , nops(%) , % , diff( p(t),t$3 ) );
```

$$\begin{aligned} Corr_Rule := & \frac{1}{24} (b - A_1)(b - A_2)(b - A_3)(b - A_4) (D^{(3)})(f)(b) \\ & - \frac{1}{24} (a - A_1)(a - A_2)(a - A_3)(a - A_4) (D^{(3)})(f)(a) - (\\ & \frac{1}{24} (b - A_2)(b - A_3)(b - A_4) + \frac{1}{24} (b - A_1)(b - A_3)(b - A_4) \\ & + \frac{1}{24} (b - A_1)(b - A_2)(b - A_4) + \frac{1}{24} (b - A_1)(b - A_2)(b - A_3))(D^{(2)})(f)(b) + (\\ & \frac{1}{24} (a - A_2)(a - A_3)(a - A_4) + \frac{1}{24} (a - A_1)(a - A_3)(a - A_4) \\ & + \frac{1}{24} (a - A_1)(a - A_2)(a - A_4) + \frac{1}{24} (a - A_1)(a - A_2)(a - A_3))(D^{(2)})(f)(a) + (\\ & \frac{1}{12} (b - A_3)(b - A_4) + \frac{1}{12} (b - A_2)(b - A_4) + \frac{1}{12} (b - A_2)(b - A_3) \\ & + \frac{1}{12} (b - A_1)(b - A_4) + \frac{1}{12} (b - A_1)(b - A_3) + \frac{1}{12} (b - A_1)(b - A_2))D(f)(b) - (\\ & \frac{1}{12} (a - A_3)(a - A_4) + \frac{1}{12} (a - A_2)(a - A_4) + \frac{1}{12} (a - A_2)(a - A_3) \\ & + \frac{1}{12} (a - A_1)(a - A_4) + \frac{1}{12} (a - A_1)(a - A_3) + \frac{1}{12} (a - A_1)(a - A_2))D(f)(a) \\ & - (b - \frac{1}{4}A_4 - \frac{1}{4}A_3 - \frac{1}{4}A_2 - \frac{1}{4}A_1)f(b) + (a - \frac{1}{4}A_4 - \frac{1}{4}A_3 - \frac{1}{4}A_2 - \frac{1}{4}A_1)f(a) \\ & + \int_a^b f(t) dt \end{aligned}$$

Impose the condition that the sum of the coefficient of each derivative at $t = a$ and $t = b$ vanishes.

```
> eq1 := coeff( Corr_Rule , D(f)(a) )
>         = -coeff( Corr_Rule , D(f)(b) );
> eq2 := coeff( Corr_Rule , (D@@2)(f)(a) )
>         = -coeff( Corr_Rule , (D@@2)(f)(b) );
> eq3 := coeff( Corr_Rule , (D@@3)(f)(a) )
>         = -coeff( Corr_Rule , (D@@3)(f)(b) );
```

$$\begin{aligned} eq1 := & -\frac{1}{12}(a - A_3)(a - A_4) - \frac{1}{12}(a - A_2)(a - A_4) - \frac{1}{12}(a - A_2)(a - A_3) \\ & - \frac{1}{12}(a - A_1)(a - A_4) - \frac{1}{12}(a - A_1)(a - A_3) - \frac{1}{12}(a - A_1)(a - A_2) = \\ & -\frac{1}{12}(b - A_3)(b - A_4) - \frac{1}{12}(b - A_2)(b - A_4) - \frac{1}{12}(b - A_2)(b - A_3) \\ & - \frac{1}{12}(b - A_1)(b - A_4) - \frac{1}{12}(b - A_1)(b - A_3) - \frac{1}{12}(b - A_1)(b - A_2) \end{aligned}$$

$$\begin{aligned} eq2 := & \frac{1}{24}(a - A_2)(a - A_3)(a - A_4) + \frac{1}{24}(a - A_1)(a - A_3)(a - A_4) \\ & + \frac{1}{24}(a - A_1)(a - A_2)(a - A_4) + \frac{1}{24}(a - A_1)(a - A_2)(a - A_3) = \\ & \frac{1}{24}(b - A_2)(b - A_3)(b - A_4) + \frac{1}{24}(b - A_1)(b - A_3)(b - A_4) \\ & + \frac{1}{24}(b - A_1)(b - A_2)(b - A_4) + \frac{1}{24}(b - A_1)(b - A_2)(b - A_3) \end{aligned}$$

$$eq3 := -\frac{1}{24}(a - A_1)(a - A_2)(a - A_3)(a - A_4) = -\frac{1}{24}(b - A_1)(b - A_2)(b - A_3)(b - A_4)$$

Solve the non-linear system to find the first three terms, A_1, A_2, A_3 of $p(t)$.

```
> solve( {'eq'|i'=1..3} , {A[i]|i'=1..3} );
```

$$\{A_3 = \%1, A_1 = a - \%1 + b, A_2 = a + b - A_4\},$$

$$\{A_3 = a + b - A_4, A_1 = a - \%1 + b, A_2 = \%1\},$$

$$\{A_3 = \%1, A_2 = a - \%1 + b, A_1 = a + b - A_4\}$$

$$\%1 := \text{RootOf}(_Z^2 + (-a - b)_Z - aA_4 + A_4^2 - bA_4 + 2ba)$$

Without loss of generality we can choose the first solution. The three solutions here appear since we have not imposed any ordering between A_1, A_2 and A_3 .

```
> Sols := allvalues( %[1] );
```

$$Sols := \{A_3 = \frac{a}{2} + \frac{b}{2} + \frac{1}{2}\%1, A_1 = \frac{a}{2} + \frac{b}{2} - \frac{1}{2}\%1, A_2 = a + b - A_4\},$$

$$\{A_3 = \frac{a}{2} + \frac{b}{2} - \frac{1}{2}\%1, A_2 = a + b - A_4, A_1 = \frac{a}{2} + \frac{b}{2} + \frac{1}{2}\%1\}$$

$$\%1 := \sqrt{a^2 - 6ba + b^2 + 4aA_4 - 4A_4^2 + 4bA_4}$$

```
> assign( %[1] );
```

Thus the new corrected trapezoidal rule appears below. Note that the second derivative has vanished. This is typical of rules of this type and it can be shown that all even derivatives vanish in "corrected" rules. The Maple command `collect` is used to collect all the f terms together,

i.e. $f(a), f(b), D(f)(a), D(f)(b)$, etc. `map(factor, ...)` applies the factor function across each of these terms.

> `map(factor, collect(Corr_Rule, f));`

$$\begin{aligned} & \frac{1}{2}(a-b)f(b) + \frac{1}{2}(a-b)f(a) + \frac{1}{12}(a-b)^2 D(f)(b) \\ & + \frac{1}{96}(-b+a-\%1)(a-A_4)(-b+a+\%1)(-b+A_4)(D^{(3)}(f))(b) \\ & + \frac{1}{96}(-b+a+\%1)(b-A_4)(-b+a-\%1)(a-A_4)(D^{(3)}(f))(a) \\ & - \frac{1}{12}(a-b)^2 D(f)(a) + \int_a^b f(t) dt \\ \%1 & := \sqrt{a^2 - 6ba + b^2 + 4aA_4 - 4A_4^2 + 4bA_4} \end{aligned}$$

The kernel is.

> `p(t);`

$$\begin{aligned} & \frac{1}{24}\left(t - \frac{a}{2} - \frac{b}{2} + \frac{1}{2}\sqrt{a^2 - 6ba + b^2 + 4aA_4 - 4A_4^2 + 4bA_4}\right)(t - a - b + A_4) \\ & \left(t - \frac{a}{2} - \frac{b}{2} - \frac{1}{2}\sqrt{a^2 - 6ba + b^2 + 4aA_4 - 4A_4^2 + 4bA_4}\right)(t - A_4) \end{aligned}$$

The term under the radical imposes a restricted domain on A_4 . We can select this term and complete the square to determine this domain.

Select the term.

> `op([2, 4, 2, 1], p(t));`

$$a^2 - 6ba + b^2 + 4aA_4 - 4A_4^2 + 4bA_4$$

Complete the square.

> `completesquare(% , A[4]);`

$$-4\left(A_4 - \frac{a}{2} - \frac{b}{2}\right)^2 + 2a^2 - 4ba + 2b^2$$

Notice that the remaining term is a perfect square, so we factor this.

> `factor(subsop(1=0, %));`

$$2(a-b)^2$$

> `fac := op(1, %%)+%;`

$$fac := -4\left(A_4 - \frac{a}{2} - \frac{b}{2}\right)^2 + 2(a-b)^2$$

It is obvious, by inspection, that the above expression is non-negative on a domain larger than $[a, b]$. We can find this domain.

> `solve(fac=0, A[4]);`

$$\frac{1}{2}a + \frac{1}{2}b - \frac{1}{2}\sqrt{2}a + \frac{1}{2}\sqrt{2}b, \frac{1}{2}a + \frac{1}{2}b + \frac{1}{2}\sqrt{2}a - \frac{1}{2}\sqrt{2}b$$

> `A4sol := map(factor, [%]);`

$$A4sol := \left[-\frac{(\sqrt{2}-1)(a-3b-2\sqrt{2}b)}{2}, \frac{(1+\sqrt{2})(a-3b+2\sqrt{2}b)}{2}\right]$$

> `A4min := A4sol[2]; A4max := A4sol[1];`

$$A_{4min} := \frac{(1 + \sqrt{2})(a - 3b + 2\sqrt{2}b)}{2}$$

$$A_{4max} := -\frac{(\sqrt{2} - 1)(a - 3b - 2\sqrt{2}b)}{2}$$

The domain of A_4 is $[A_{4min}, A_{4max}]$ where $A_{4min} < a$ and $b < A_{4max}$. We can now inspect the nature of the kernel over the unit interval.

```
> A4min01 := expand(subs( a=0,b=1, A4min ));
```


$$A_{4min01} := \frac{1}{2} - \frac{\sqrt{2}}{2}$$

```
> A4max01 := expand(subs( a=0,b=1, A4max ));
```

$$A_{4max01} := \frac{1}{2} + \frac{\sqrt{2}}{2}$$

The following animated and 3D plots reveal that p generally changes sign twice as A_4 increases from A_{4min} to A_{4max} . We already know the roots of p , they are the terms A_1 , A_2 , A_3 and A_4 and so to determine the sign of the kernel for any t we require knowledge of how the A_i , $i = 1..4$ relate to each other and the interval $[a, b]$.

```
> animate( plot , [ subs(a=0,b=1,p(t)) , t=0..1 ],
> A[4]=A4min01..A4max01 , frames=60 );
```




3D plot of the kernel.

```
> plot1 := contourplot( subs(a=0,b=1,p(t)) , t=0..1 ,
> A[4]=A4min01..A4max01 , contours = 15 );

> plot2 := plot3d( subs(a=0,b=1,p(t)) , t=0..1 ,
> A[4]=A4min01..A4max01 , grid=[30,30] , axes=boxed );

> ptrans := transform((x,y) -> [x,y,-0.006]):


> display( {ptrans(plot1),plot2},orientation=[-135,65]);
```



The following plot reveals that the kernel coefficients change ordering across four regions. They are:

- $A[4] < a < A[1] < A[3] < b < A[2]$, for $A4min < A[4] < a$
- $A[1] < a < A[4] < A[2] < b < A[3]$, for $a < A[4] < (a+b)/2$
- $A[1] < a < A[2] < A[4] < b < A[3]$, for $(a+b)/2 < A[4] < b$
- $A[2] < a < A[1] < A[3] < b < A[4]$, for $b < A[4] < A4max$

```
> plot( subs(a=0,b=1,[A[i]$i=1..4]) , A[4]=A4min01..A4max01 ,
> thickness=4,color=[blue,black,yellow,red] );
```



A more descriptive plot than the Maple graphic above is shown in Figure 5.1.

The upper bound Ib , which is the integral of the positive kernel, requires four separate calculations according to the regions defined above.

```
A[4] < a < A[1] < A[3] < b < A[2], for A4min < A[4] < a
> Ib1 := factor( -int( p(t) , t=a..A[1] )
> + int( p(t) , t=A[1]..A[3] )
> - int( p(t) , t=A[3]..b ) );
A[1] < a < A[4] < A[2] < b < A[3], for a < A[4] < (a+b)/2
> Ib2 := factor( -int( p(t) , t=a..A[4] )
> + int( p(t) , t=A[4]..A[2] ) - int( p(t) , t=A[2]..b ) );
A[1] < a < A[2] < A[4] < b < A[3], for (a+b)/2 < A[4] < b
```




Figure 5.1: A graphic showing the ordering of A_1, A_2, A_3, A_4 in regions R_1, R_2, R_3, R_4 .

```
> Ib3 := factor( -int( p(t) , t=a..A[2] )
> + int( p(t) , t=A[2]..A[4] ) - int( p(t) , t=A[4]..b ) ):
A[2]<a<A[1] <A[3]<b< A[4], for b<A[4]<A4max
```


```
> Ib4 := Ib1:
```

Hence, we can define Ib in a piecewise fashion.

```
> Ib := piecewise( A[4]<a , Ib1 , A[4]<(a+b)/2 , Ib2 ,
> A[4] < b , Ib3 , A[4]<A4max , Ib4 , 0 ):
```

As a function of A_4 , the upper bound reveals four minimum points. We will explore the first minimum point and the new corrected rule resulting from evaluating A_4 at this point. The results are the same for all four minima.

```
> plot( subs(a=0,b=1,Ib) , A[4]=A4min01..A4max01 );
```



The first minimum point is a stationary point of $Ib1$.

```

> assume( b>a );
> solve(diff( Ib1,A[4] ) , A[4]);
 $b^{\sim}, \frac{1}{2}b^{\sim} + \frac{1}{2}a^{\sim} - \frac{1}{4}\sqrt{7}a^{\sim} + \frac{1}{4}\sqrt{7}b^{\sim}, \frac{1}{2}b^{\sim} + \frac{1}{2}a^{\sim} + \frac{1}{4}\sqrt{7}a^{\sim} - \frac{1}{4}\sqrt{7}b^{\sim}, \frac{a^{\sim}}{2} + \frac{b^{\sim}}{2}, a^{\sim}$ 
Select the correct point since for Ib1 we must have  $A_{4min} < A_4$  and  $A_4 < a$ .
> sol1 :=[ % ][3];
 $sol1 := \frac{1}{2}b^{\sim} + \frac{1}{2}a^{\sim} + \frac{1}{4}\sqrt{7}a^{\sim} - \frac{1}{4}\sqrt{7}b^{\sim}$ 

```

Hence the upper bound is

```

> factor( subs(A[4]=sol1 , Ib1 ) );

$$-\frac{5(a^{\sim} - b^{\sim})^5}{6144}$$


```

and the new corrected trapezoidal rule is

```

> map( factor@expand , subs(A[4]=sol1,Corr_Rule) );

$$-\frac{3}{2048}(D^{(3)}(f)(b^{\sim})(a^{\sim} - b^{\sim})^4 + \frac{3}{2048}(D^{(3)}(f)(a^{\sim})(a^{\sim} - b^{\sim})^4 + \frac{1}{12}D(f)(b^{\sim})(a^{\sim} - b^{\sim})^2$$


$$- \frac{1}{12}D(f)(a^{\sim})(a^{\sim} - b^{\sim})^2 + \frac{1}{2}f(b^{\sim})(a^{\sim} - b^{\sim}) + \frac{1}{2}f(a^{\sim})(a^{\sim} - b^{\sim}) + \int_{a^{\sim}}^{b^{\sim}} f(t) dt$$


```

When compared to the generic corrected trapezoidal rule

$$\int_a^b f(x) dx = \frac{(b-a)f(a)}{2} + \frac{(b-a)f(b)}{2} + \frac{(b-a)^2 D(f)(a)}{12} - \frac{(b-a)^2 D(f)(b)}{12} + E$$

where

$$|E| \leq \frac{\max(|(D^{(4)}(f))|) (b - a)^5}{720}$$

we can see that the new rule has two added third derivative function evaluations (which will cancel on composition) and an upper bound coefficient of 5/6144. This is approximately 1.7 times better.

```

> evalf( (1/720) / (5/6144) );
1.706666667

```

In practice, if we had knowledge of the 3rd derivative on the boundary then we would use the 6th order corrected trapezoidal rule

$$\left| -\frac{(a - b)^4 (D^{(3)}(f)(b))}{720} + \frac{(a - b)^4 (D^{(3)}(f)(a))}{720} + \frac{(a - b)^2 D(f)(b)}{12} - \frac{(a - b)^2 D(f)(a)}{12} + \frac{(a - b) f(b)}{2} + \frac{(a - b) f(a)}{2} + \int_a^b f(t) dt \right| \leq \frac{\max(|(D^{(6)}(f))|) (b - a)^7}{30240}.$$

Nevertheless, we have shown that Maple is a useful tool for exploring quadrature rules and their error bounds.

6. MODIFIED SIMPSON'S RULE

In this section, we will use the Maple computer algebra system to derive a modified Simpson rule that was recently established. This section is an actual Maple worksheet called **mod_simp_6.mws** which you can download from <http://www.roumeliotis.com.au/john/maple>.

Ujević and Roberts (2004) recently established the following modified Simpson type rule:

$$\int_a^b f(x)dx = \frac{(b-a)(7f(a) + 16f(\frac{a+b}{2}) + 7f(b))}{30} - \frac{(b-a)^2(D(f)(b) - D(f)(a))}{60} + E$$

where $|E| \leq \frac{\max(|(D^{(6)}(f))|)(b-a)^7}{604800}$. We can see that the weights are different to those of the well known Simpson inequality and that the end-point derivative correction produces a rule of order 6.

Using Maple, we will recapture this result by employing a split 6th order Peano kernel. In order to eliminate the presence of higher derivatives at the end-points, the kernel will require the correct smoothness at $t=a,b$.

```
> restart;
```

```
> with( student );
```

The $(t-a)^4$ and $(t-b)^4$ terms will ensure no derivatives higher than one exist at the end-points. Since this will be a sixth order rule, the kernel needs to be a sixth order polynomial. Thus we have four free parameters $c0, c1, c2, c3$.

```
> p1 := 1/6!*(t-a)^4*(t^2+c0*t+c1);
```

```
> p2 := 1/6!*(t-b)^4*(t^2+c2*t+c3);
```

$$p1 := \frac{(t-a)^4(t^2 + c0t + c1)}{720}$$

$$p2 := \frac{(t-b)^4(t^2 + c2t + c3)}{720}$$

```
> I1 := Int( p1 * (D@@6)(f)(t) , t=a..(a+b)/2 );
```

```
> I2 := Int( p2 * (D@@6)(f)(t) , t=(a+b)/2..b );
```

$$I1 := \int_a^{\frac{a}{2} + \frac{b}{2}} \frac{1}{720} (t-a)^4 (t^2 + c0t + c1) (D^{(6)}(f)(t) dt$$

$$I2 := \int_{\frac{a}{2} + \frac{b}{2}}^b \frac{1}{720} (t-b)^4 (t^2 + c2t + c3) (D^{(6)}(f)(t) dt$$

Integrate by parts six times.

```
> I1 := intparts( I1 , p1 );
```

```
> I2 := intparts( I2 , p2 );
```

```
> for i from 1 to 5 do
```

```
> I1 := applyop( intparts , nops(I1) , I1 , diff(p1,t$i) );
```

```
> I2 := applyop( intparts , nops(I2) , I2 , diff(p2,t$i) );
```

```
> end do;
```

We can see that the integration rule contains sampling up to order one at the endpoints and sampling up to order 5 at the midpoint. The midpoint sampling is due to the fact that the Peano kernel is discontinuous here.

```
> I_MS := map( factor , collect( combine( I1+I2 ) , f ) );
```

$$\begin{aligned}
 I_{MS} := & \frac{1}{23040} (a-b)^4 (-c_2 a + c_0 a - 2 c_3 + 2 c_1 + c_0 b - c_2 b) (D^{(5)})(f)\left(\frac{a}{2} + \frac{b}{2}\right) + \\
 & \frac{1}{11520} (a-b)^3 \\
 & (4 a^2 + 3 c_0 a + 5 c_2 a + 8 a b + 8 c_1 + 8 c_3 + 5 c_0 b + 4 b^2 + 3 c_2 b) \\
 & (D^{(4)})(f)\left(\frac{a}{2} + \frac{b}{2}\right) - \frac{1}{1440} \\
 & (a-b)^2 (4 a^2 - c_0 a + 5 c_2 a - 6 c_1 - 5 c_0 b + 6 c_3 + c_2 b - 4 b^2) (D^{(3)})(f)\left(\frac{a}{2} + \frac{b}{2}\right) \\
 & + \frac{1}{240} \\
 & (a-b) (4 a^2 - c_0 a + 5 c_2 a + 5 c_0 b + 4 c_3 - c_2 b + 4 b^2 + 4 c_1) (D^{(2)})(f)\left(\frac{a}{2} + \frac{b}{2}\right) \\
 & - \frac{1}{60} (5 c_2 a - 3 c_2 b + 2 c_3 + 8 a^2 - 8 b^2 + 3 c_0 a - 5 c_0 b - 2 c_1) D(f)\left(\frac{a}{2} + \frac{b}{2}\right) \\
 & - \frac{1}{30} (a^2 + c_0 a + c_1) D(f)(a) + \frac{1}{30} (b^2 + c_2 b + c_3) D(f)(b) \\
 & + \frac{1}{6} (4 a - 4 b - c_0 + c_2) f\left(\frac{a}{2} + \frac{b}{2}\right) + \int_a^b f(t) dt - \frac{1}{6} (2 b + c_2) f(b) \\
 & + \frac{1}{6} (2 a + c_0) f(a)
 \end{aligned}$$

Choose c_0, c_1, c_2, c_3 so that all sampling of derivatives at the midpoint vanish.

```

> eq1 := map( factor , collect( coeff( I_MS ,
> D(f)((a+b)/2) ) , [c0,c1,c2,c3] ) ) = 0;

```

$$eq1 := -\frac{(3a-5b)c_0}{60} + \frac{c_1}{30} - \frac{(5a-3b)c_2}{60} + \frac{2b^2}{15} - \frac{c_3}{30} - \frac{2a^2}{15} = 0$$

```

> eq2 := collect( coeff( I_MS ,
> (D@@2)(f)((a+b)/2) ) , [c0,c1,c2,c3] ) = 0;

```

$$\begin{aligned}
 eq2 := & \frac{(a-b)(-a+5b)c_0}{240} + \left(\frac{a}{60} - \frac{b}{60}\right)c_1 + \frac{(a-b)(5a-b)c_2}{240} + \left(\frac{a}{60} - \frac{b}{60}\right)c_3 \\
 & + \frac{(a-b)(4a^2+4b^2)}{240} = 0
 \end{aligned}$$

```

> eq3 := collect( coeff( I_MS ,
> (D@@3)(f)((a+b)/2) ) , [c0,c1,c2,c3] ) = 0;

```

$$\begin{aligned}
 eq3 := & -\frac{(a-b)^2(-a-5b)c_0}{1440} + \frac{(a-b)^2c_1}{240} - \frac{(a-b)^2(5a+b)c_2}{1440} - \frac{(a-b)^2c_3}{240} \\
 & - \frac{(a-b)^2(4a^2-4b^2)}{1440} = 0
 \end{aligned}$$

```

> eq4 := collect( coeff( I_MS ,
> (D@@4)(f)((a+b)/2) ) , [c0,c1,c2,c3] ) = 0;

```

$$\begin{aligned}
 eq4 := & \frac{(a-b)^3(3a+5b)c_0}{11520} + \frac{(a-b)^3c_1}{1440} + \frac{(a-b)^3(5a+3b)c_2}{11520} + \frac{(a-b)^3c_3}{1440} \\
 & + \frac{(a-b)^3(4a^2+8ab+4b^2)}{11520} = 0
 \end{aligned}$$

```
> eq5 := collect( coeff( I_MS ,
> (D@@5)(f)((a+b)/2) ) , [c0,c1,c2,c3] ) = 0;
eq5 := \frac{(a-b)^4(a+b)c0}{23040} + \frac{(a-b)^4c1}{11520} + \frac{(a-b)^4(-a-b)c2}{23040} - \frac{(a-b)^4c3}{11520} = 0
```

Solve to obtain the coefficients

```
> solve( {eq1,eq2,eq3,eq4} , {c0,c1,c2,c3} );
{c2 = -\frac{7a}{5} - \frac{3b}{5}, c3 = \frac{1}{2}a^2 + \frac{2}{5}ab + \frac{1}{10}b^2, c0 = -\frac{3a}{5} - \frac{7b}{5}, c1 = \frac{1}{10}a^2 + \frac{2}{5}ab + \frac{1}{2}b^2}
> assign( % );
```

Hence, the integration rule is:

```
> I_MS := map( factor , I_MS );
```

$$I_{MS} := -\frac{1}{60}(a-b)^2 D(f)(a) + \frac{1}{60}(a-b)^2 D(f)(b) + \frac{8}{15}(a-b)f\left(\frac{a}{2} + \frac{b}{2}\right) + \int_a^b f(t) dt$$

$$+ \frac{7}{30}(a-b)f(b) + \frac{7}{30}(a-b)f(a)$$

and the split kernel is

```
> p1 := factor( p1 );
p1 := \frac{(-t+a)^4(10t^2-6ta-14tb+a^2+4ab+5b^2)}{7200}
```

```
> p2 := factor( p2 );
p2 := \frac{(t-b)^4(10t^2-14ta-6tb+5a^2+4ab+b^2)}{7200}
```

Completing the square in both $p1$ and $p2$ shows that the kernel is positive

```
> p1 := applyop( completesquare , 3 , p1 , t );
p1 := \frac{(-t+a)^4(10(t-\frac{3a}{10}-\frac{7b}{10})^2+\frac{a^2}{10}-\frac{ab}{5}+\frac{b^2}{10})}{7200}
> p1 := subsop( 3 = op([3,1],p1) + factor(op(3,p1)
> - op([3,1],p1)) , p1 );
```

$$p1 := \frac{(-t+a)^4(10(t-\frac{3a}{10}-\frac{7b}{10})^2+\frac{(a-b)^2}{10})}{7200}$$

```
> p2 := applyop( completesquare , 3 , p2 , t );
```

$$p2 := \frac{(t-b)^4(10(t-\frac{7a}{10}-\frac{3b}{10})^2+\frac{a^2}{10}-\frac{ab}{5}+\frac{b^2}{10})}{7200}$$

```
> p2 := subsop( 3 = op([3,1],p2) + factor(op(3,p2)
> - op([3,1],p2)) , p2 );
```

$$p2 := \frac{(t-b)^4(10(t-\frac{7a}{10}-\frac{3b}{10})^2+\frac{(a-b)^2}{10})}{7200}$$

Thus the upper bound is calculated directly from the integral of the kernel

```
> factor( int( p1 , t=a..(a+b)/2 )
> + int( p2 , t=(a+b)/2..b ) );
-\frac{(a-b)^7}{604800}
```


We have used this Maple session to show how one may apply a computer algebra system such as Maple to easily recapture known results and extend to new integration rules.

7. OTHER RESULTS

In this section we briefly state other results obtained using the technique outlined in Sections 2 – 6. For each result, we state the general Peano kernel complete with free parameters and an outline of the steps taken to produce the integral inequality.

- (1) **Fourth Order Modified Simpson Inequality.** The kernel below will produce a fourth order rule with sampling of all derivatives at the discontinuous point x and only sampling of f at the endpoints $t = a, b$.

$$p(t) = \begin{cases} \frac{1}{4!}(t - a)^3(t - a_0), & t \in [a, x), \\ \frac{1}{6!}(t - b)^3(t - b_0), & t \in [x, b]. \end{cases}$$

Given this kernel the motivation is to produce an optimal rule. That is, minimize the upper bound using similar steps as shown in Section 5. The result is

$$\left| \int_a^b f(t) dt - \frac{b-a}{16} \left(2^{3/4} f(a) + 2(2 - 2^{1/4}) \left(\sqrt{2} + 2 \cdot 2^{1/4} + 4 \right) f\left(\frac{a+b}{2}\right) + \frac{(b-a)^2}{12} (8 - 3 \cdot 2^{3/4}) f''\left(\frac{a+b}{2}\right) + 2^{3/4} f(b) \right) \right| \leq \frac{2 - 2^{3/4}}{3840} (b-a)^5 \|f^{(4)}\|_\infty$$

- (2) **Sixth Order Modified Simpson Inequality.** Multiply each branch of the well known Simpson kernel with an arbitrary quadratic polynomial.

$$p(t) = \begin{cases} \frac{1}{6!}(t - a)^3(t - \frac{a+2b}{3})(t^2 + a_1t + a_0), & t \in [a, \frac{a+b}{2}) \\ \frac{1}{6!}(t - b)^3(t - \frac{2a+b}{3})(t^2 + b_1t + b_0), & t \in [\frac{a+b}{2}, b] \end{cases}$$

Find the coefficients so that rule is corrected, that is end-point derivatives are equal and opposite, and as many midpoint evaluations vanish as possible. The result is

$$(7.1) \quad \left| \int_a^b f(t) dt - \left[\frac{b-a}{18} \left(5f(a) + 8f\left(\frac{a+b}{2}\right) + 5f(b) \right) + \frac{(b-a)^2}{45} (f'(a) - f'(b)) - \frac{(b-a)^3}{180} f''\left(\frac{a+b}{2}\right) \right] \right| \leq \frac{19}{7257600} (b-a)^7 \|f^{(6)}\|_\infty$$

- (3) **Eighth Order Modified Simpson Inequality - I.** Multiply each branch of the well known Simpson kernel with an arbitrary quartic polynomial.

$$p(t) = \begin{cases} \frac{1}{8!}(t - a)^3(t - \frac{a+2b}{3})(t^4 + a_3t^3 + a_2t^2 + a_1t + a_0), & t \in [a, \frac{a+b}{2}) \\ \frac{1}{8!}(t - b)^3(t - \frac{2a+b}{3})(t^4 + b_3t^3 + b_2t^2 + b_1t + b_0), & t \in [\frac{a+b}{2}, b] \end{cases}$$

Find the coefficients so that rule is corrected, that is end-point derivatives are equal and opposite, and as many midpoint evaluations vanish as possible. The result is

$$(7.2) \quad \left| \int_a^b f(t) dt - \left[\frac{b-a}{294} \left(59f(a) + 176f\left(\frac{a+b}{2}\right) + 59f(b) \right) \right. \right. \\ \left. \left. + \frac{55}{4166}(b-a)^2(f'(a) - f'(b)) + \frac{5}{1029}(b-a)^3 f''\left(\frac{a+b}{2}\right) + \frac{(b-a)^4}{30870}(f'''(b) - f'''(a)) \right] \right| \\ \leq \frac{71}{19914854400}(b-a)^9 \|f^{(8)}\|_\infty$$

- (4) **Eighth Order Modified Simpson Inequality - II.** In an effort to eliminate the $f''((a+b)/2)$ term in (7.2), we replace the kernel above with the following

$$p(t) = \begin{cases} \frac{1}{8!}(t-a)^4(t^4 + a_3t^3 + a_2t^2 + a_1t + a_0), & t \in [a, \frac{a+b}{2}] \\ \frac{1}{8!}(t-b)^4(t^4 + b_3t^3 + b_2t^2 + b_1t + b_0), & t \in [\frac{a+b}{2}, b] \end{cases}$$

This kernel eliminates sampling of $f^{(i)}$, $i = 8, 7, 6, 5$ at $t = a, b$. Solve for coefficients so that rule is corrected and as many midpoint evaluations vanish as possible. The integration rule is

$$(7.3) \quad \left| \int_a^b f(t) dt - \left[\frac{b-a}{126} \left(31f(a) + 64f\left(\frac{a+b}{2}\right) + 31f(b) \right) \right. \right. \\ \left. \left. + \frac{5}{252}(b-a)^2(f'(a) - f'(b)) + \frac{(b-a)^4}{15120}(f'''(b) - f'''(a)) \right] \right| \\ \leq \frac{(b-a)^9}{101606400} \|f^{(8)}\|_\infty$$

- (5) **Order Ten Corrected Trapezoidal Rule** Multiply the corrected trapezoidal kernel of Section 4 (see 3rd last equation in this section) with an arbitrary sixth order polynomial.

$$p(t) = \frac{1}{10!}(t-a)^2(t-b)^2(t^6 + a_5t^5 + a_4t^4 + a_3t^3 + a_2t^2 + a_1t + a_0)$$

Solve for coefficients so that rule is corrected, that is all end-point derivatives are equal and opposite in sign. The resultant integral inequality is:

$$(7.4) \quad \left| \int_a^b f(t) dt - \left[\frac{b-a}{2}(f(a) + f(b)) + \frac{(b-a)^2}{12}(f'(a) - f'(b)) \right. \right. \\ \left. \left. + \frac{(b-a)^4}{720}(f'''(b) - f'''(a)) + \frac{(b-a)^6}{30240}(f^{(5)}(a) - f^{(5)}(b)) + \frac{(b-a)^8}{1209600}(f^{(7)}(b) - f^{(7)}(a)) \right] \right| \\ \leq \frac{(b-a)^{11}}{47900160} \|f^{(10)}\|_\infty.$$

REFERENCES

- [1] N.S. BARNETT, P. CERONE, S.S. DRAGOMIR, J. ROUMELIOTIS, AND A. SOFO, A survey on Ostrowski type inequalities for twice differentiable mappings and applications,

- In Y. J. Cho, S.K. Kim, and S.S. Dragomir (Eds), *Inequality Theory & Applications*, Vol. 1, Nova Sciences Publishers Inc., 2001, pages 35–88.
- [2] P. CERONE, Three point rules in numerical integration, *Nonlinear Analysis*, **47** (2001), 2341–2352.
- [3] P. CERONE, On perturbed trapezoidal and midpoint rules, *Korean J. of Comput. & Appl. Math.*, **2**(2002), 423–435.
- [4] P. CERONE, On an identity for the Chebychev functional and some ramifications, *J. Inequal. Pure & Appl. Math.*, **3**(1) (2002), Art. 4. [ONLINE: http://jipam.vu.edu.au/v3n1/034_01.html].
- [5] P. CERONE, Generalised trapezoidal rules with error involving bounds of the n th derivative, *Math. Ineq. & Appl.*, **5**(3) (2002), 451–462. [ONLINE: <http://www.mia-journal.com/contents.asp?number=19>].
- [6] P. CERONE, On relationships between Ostrowski, trapezoidal and Chebychev identities and inequalities, *Soochow J. of Math.*, **28**(3) (2002), 311–328.
- [7] S.S. DRAGOMIR, On Simpson’s quadrature formula for differentiable mappings whose derivatives belong to l_p -spaces and applications, *J. KSIAM*, **2**(2) (1998), 57–65.
- [8] S.S. DRAGOMIR AND S. WANG, Applications of Ostrowski inequality to the estimation of error bounds for some special means and some numerical quadrature rules, *Appl. Math. Lett.*, **11** (1998), 105–109.
- [9] S.S. DRAGOMIR, On Simpson’s quadrature formula for mappings of bounded variation and applications, *Tamkang J. Math.*, **30**(1) (1999), 53–58.
- [10] S.S. DRAGOMIR, P. CERONE AND J. ROUMELIOTIS, A new generalisation of Ostrowski’s integral inequality for mappings whose derivatives are bounded and applications in numerical integration and for special means, *Appl. Math. Lett.*, **13** (2000), 19–25.
- [11] S.S. DRAGOMIR AND Th.M. RASSIAS (Eds.), *Ostrowski Type Inequalities and Applications in Numerical Integration*, Kluwer Academic, Dordrecht, 2002.
- [12] S.S. DRAGOMIR, J. PEČARIĆ AND S. WANG, The unified treatment of trapezoid, Simpson, and Ostrowski type inequality for monotonic mappings and applications, *Math. Comput. Modelling*, **31** (2000), 61–70.
- [13] M.B. MONAGAN, K.O. GEDDES, K.M. HEAL, G. LABAHN, S.M. VORKOETTER, J. MCCARRON, AND P. DEMARCO, *Maple 8 Introductory Programming Guide*, Waterloo Maple Inc., Waterloo, 2002.
- [14] J. PEČARIĆ AND S. VAROŠANEC, A note on Simpson’s inequality for functions of bounded variation, *Tamkang J. Math.*, **31**(3) (2000) 239–242.
- [15] J. PEČARIĆ AND S. VAROŠANEC, Harmonic polynomials and generalizations of Ostrowski inequality with applications in numerical integration. *Nonlinear Analysis*, **42** (2001), 2365–2374.
- [16] J. PEČARIĆ AND S. VAROŠANEC, Simpson’s formula for functions whose derivatives belong to l_p spaces, *Appl. Math. Lett.*, **14**(2) (2001), 131–135.
- [17] N. UJEVIĆ AND A.J. ROBERTS, A corrected quadrature formula and applications, *ANZIAM J.*, **45**(E) (2004), E41–E56. [ONLINE: <http://anziamj.austms.org.au/V45/E051/home.html>].